

Problem 1

- 10 potential sites, build 4 max
- each site has fixed capacity
- each resident needs access to one site
- residential areas $i \in \{1 \dots 50\}$, each w/ r_i residents
- sites $j \in \{1 \dots 10\}$, each w/ capacity c_j
- d_{ij} = distance from i to j = $|x_i - x_j| + |y_i - y_j|$

(a) Variables: $s_j = \begin{cases} 1 & \text{if site } j \text{ is open} \\ 0 & \text{otherwise} \end{cases} \quad \forall j \in [10]$ (indicates which sites to open)

v_{ij} = number of residents from area i served by site $j \quad \forall i \in [50], j \in [10]$

Constraints: build 4 sites max: $\sum_{j=1}^{10} s_j \leq 4$

total # residents coming from residential area i must equal # residents in i (every resident assigned to exactly 1 site)

$$\sum_{j=1}^{10} v_{ij} = r_i \quad \forall i \in [50]$$

total # residents served by site j must be within site j 's capacity:

$$\sum_{i=1}^{50} v_{ij} \leq c_j \quad \forall j \in [10]$$

define distance (not nonlinear because d_{ij} not a variable - data is given)

$d_{ij} = |x_i - x_j| + |y_i - y_j| \quad \forall i \in [50], j \in [10]$
(this is not an actual constraint, it's a matrix of constants that can be defined when the data is loaded)

if residents from i are served by site j , then j must be built:

$$v_{ij} \leq s_j \cdot c_j \quad \forall i \in [50], j \in [10]$$

binary variable: $s_j \in \{0, 1\} \quad \forall j \in [10]$

positive integer: $v_{ij} \in \mathbb{Z} \geq 0 \quad \forall i \in [50], j \in [10]$

Objective: minimize total distance residents must travel to vaccination clinics

$$\min_{s, v} \sum_{i=1}^{50} \sum_{j=1}^{10} d_{ij} v_{ij}$$

b) average distance : 1.858
 largest distance: 3.731

load in data

```
1 areas = CSV.read("P1_areas.csv", DataFrame, header=["x", "y", "num_res"], skipto=2)
2 clinics = CSV.read("P1_clinics.csv", DataFrame, header=["x", "y", "capacity"], skipto=2)
```

...

```
1 r = areas.num_res;
2 C = clinics.capacity;
3 x_res = areas.x;
4 y_res = areas.y;
5 x_clin = clinics.x;
6 y_clin = clinics.y;
```

```
1 # distance matrix: manhattan distance from res area i to clinic j
2 d = [abs(x_res[i]-x_clin[j]) + abs(y_res[i]-y_clin[j]) for i=1:nrow(areas), j=1:nrow(clinics)];
```

define the model

```
1 m1 = Model(Gurobi.Optimizer);
```

Set parameter Username

Academic license - for non-commercial use only - expires 2022-09-06

```
1 # variables
2 # s[j] indicates whether or not site j is opened
3 @variable(m1, s[1:nrow(clinics)], Bin) # nrow(clinics) = number of clinics
4 # v[i,j] represents how many residents from area i go to site j
5 @variable(m1, v[1:nrow(areas), 1:nrow(clinics)] >= 0, Int) # nrow(areas) = number of residential areas
```

...

```
1 # constraints
2 # build 4 sites max
3 @constraint(m1, sum(s[j] for j=1:nrow(clinics)) <= 4)
4 # total num residents coming from res area i must be equal to the number of residents
5 # (equivalent to every resident must be assigned to exactly 1 site)
6 @constraint(m1, [i=1:nrow(areas)], sum(v[i,j] for j=1:nrow(clinics)) == r[i])
7 # total num residents served by site j must be within site j's capacity
8 @constraint(m1, [j=1:nrow(clinics)], sum(v[i,j] for i=1:nrow(areas)) <= C[j])
9 # if residents from i are served by site j, then j must be built
10 @constraint(m1, [i=1:nrow(areas), j=1:nrow(clinics)], v[i,j] <= C[j]*s[j])
```

...

```
1 # objective: min total distance residents travel to clinics
2 @objective(m1, Min, sum(sum(d[i,j]*v[i,j] for j=1:nrow(clinics)) for i=1:nrow(areas)));
```

optimize the model and evaluate solution

```
1 optimize!(m1)
```

...

```
1 # see which clinics got built and min total distance
2 @show [value(s[j]) for j=1:nrow(clinics)]
3 @show objective_value(m1)
```

```
[value(s[j]) for j = 1:nrow(clinics)] = [-0.0, 1.0, -0.0, 0.0, 1.0, 1.0, -0.0, 0.0, -0.0, 1.0]
objective_value(m1) = 36763.43319948599
```

```
1 36763.43319948599
```

```
1 # v_opt is a matrix indicating how many people from each res area go to each site
2 v_opt = [value(v[i,j]) for i=1:nrow(areas), j=1:nrow(clinics)]
3
4 # find which residents i go to which clinics j
5 # (the indices for values >0 in v_opt)
6 # this is a vector of i,j pairs indicating res->clin assignments
7 res_to_clin = findall(i->(i>0), v_opt)
```

...

```
1 # index matrix d by res_to_clin to get a vector of distances that
2 # each res area must travel to get to its assigned clinic
3 distances = d[res_to_clin]
4
5 # v_opt[res_to_clin] is vector of num people going from each
6 # res area to each clinic that was built
7 # the dot prod of distances and v_opt[res_to_clin] is the total distance traveled
8 total_distance = sum(v_opt[res_to_clin][i]*distances[i] for i=1:length(distances))
9 # this is also equal to the objective value
```

```
1 36763.43319948599
```

```
1 # find max distance
2 print("max distance: ", maximum(distances))
3 # find average distance
4 print("\navg distance: ", total_distance/sum(r))
```

```
max distance: 3.7314384460000003
avg distance: 1.8578650292847174
```

plot of locations of residents & chosen clinics:

plot locations of clinics and residents

```
1 s_opt = [value(s[j]) for j=1:nrow(clinics)]
2 # find which clinics were built (the indices for values >0 in s_opt)
3 clinics_built_idx = findall(i->(i>0), s_opt)
```

4-element Vector{Int64}:


```
2
5
6
10
```

```
1 # subset the clinics df to just data on the ones that were built
2 clinics_built = clinics[clinics_built_idx, :]
```

4 rows × 3 columns

	x	y	capacity
	Float64	Float64	Int64
1	1.56519	1.67896	7867
2	4.52658	7.86786	13338
3	1.5845	7.02909	8664
4	4.69616	3.43461	12670

```
1 scatter(areas.x, areas.y, label="residential areas")
2 scatter!(clinics_built.x, clinics_built.y, label="clinics")
```



c) add constraint $\sum_{j=1}^{10} F_j s_j \leq B$ (total cost of opening clinics must be within B)

Problem 2

- $n=17$ patients, each needs 15 fractions
- can be proton and/or photon: photon = $15-p$ ($p=\#$ proton fractions)
- score for each patient $i \in (n)$: $BED_i(p, 15-p)$ given for each $0 \leq p \leq 15$
- max of C proton fractions (sum of p across all patients $\leq C$)

a) Variables: $p_{ij} = \begin{cases} 1 & \text{if patient } i \text{ receives } j-1 \text{ proton fractions } \forall i \in (n), j \in \{1, \dots, 16\} \\ 0 & \text{otherwise} \end{cases}$

(proton fractions range from 0 to 15. add 1 to get index of matrix)

Constraints: total proton fractions given across all patients can't exceed C :

$$\sum_{i=1}^{17} \sum_{j=1}^{16} p_{ij} \cdot (j-1) \leq C$$

if $p_{ij}=1$, add $(j-1)$ proton fractions to the sum - $(j-1)=\#$ proton fractions

each patient should be assigned to only one proton fraction amount:

$$\sum_{j=1}^{16} p_{ij} = 1 \quad \forall i \in [17]$$

p_{ij} binary: $p_{ij} \in \{0,1\} \quad \forall i \in (n), j \in \{0 \dots 15\}$

Objective: maximize total BED scores across all patients:

$$\max_p \sum_{i=1}^{17} \sum_{j=0}^{15} p_{ij} \cdot BED_i(j, 15-j)$$

b)

load in data

```

1 # 17x16 df
2 # rows are patient i and cols are j-1 proton fractions (idx from 1:16)
3 bed = CSV.read("outcome_data.csv", DataFrame, header=false)

```

...

solve for C = [20, 40, 50]

```

1 C = [20, 40, 50]
2 max_scores = []
3
4 for x in C
5     m2 = Model(Gurobi.Optimizer);
6
7     # variables
8     # p[i,j] indicates whether patient i receives j proton fractions
9     @variable(m2, p[1:17, 1:16], Bin);
10
11     # constraints
12     # total can't exceed max capacity of x proton fractions
13     @constraint(m2, sum(sum(p[i,j]*(j-1) for i=1:17) for j=1:16) <= x);
14     # each patient is assigned exactly 1 proton fraction amount
15     @constraint(m2, [i=1:17], sum(p[i,j] for j=1:16) == 1);
16
17     # objective: maximize total BED score
18     # bed[i,j]: BED score for patient i receiving j-1 proton fractions
19     @objective(m2, Max, sum(sum(bed[i,j]*p[i,j] for i=1:17) for j=1:16));
20
21     optimize!(m2);
22
23     append!(max_scores, objective_value(m2))
24
25     p_opt = [value(p[i,j]) for i=1:17, j=1:16]
26     # get i,j pairs of patients i who received j-1 proton fractions
27     p_assigned = findall(i->(i>0), p_opt)
28     # df displaying number of proton fractions assigned to each patient
29     p_df = DataFrame(Patient=[p_assigned[i][1] for i=1:17],
30                     Proton_Fractions=[p_assigned[i][2]-1 for i=1:17])
31     sort!(p_df, [:Patient])
32     @show(p_df)
33 end

```

...

```

1 for i=1:3
2     println("Max BED score for C = ", C[i], ": ", max_scores[i])
3 end

```

Max BED score for C = 20: 1437.215

Max BED score for C = 40: 1580.373

Max BED score for C = 50: 1619.601

C=20

Patient Int64	Proton_Fractions Int64
1	2
2	0
3	0
4	0
5	0
6	0
7	1
8	2
9	9
10	0
11	4
12	0
13	0
14	0
15	2
16	0
17	0

C=40

Patient Int64	Proton_Fractions Int64
1	2
2	0
3	1
4	0
5	0
6	0
7	3
8	5
9	9
10	1
11	5
12	3
13	0
14	0
15	7
16	4
17	0

C=50

Patient Int64	Proton_Fractions Int64
1	2
2	3
3	1
4	0
5	0
6	0
7	6
8	5
9	9
10	4
11	5
12	4
13	0
14	0
15	7
16	4
17	0